
ABSTRACT 
 
When addressing safety issues of concern, it is often necessary to provide narratives on specific patients. Indeed, the ICH-E3 
guideline calls for provision of narratives of deaths, other serious adverse events, and certain other significant adverse events 
judged to be of special interest. To facilitate for this, it is useful to provide programmed assisted narratives (PANs) in the form of 
targeted patient profiles. PANs are also useful for internal safety review, safety monitoring, and data integrity checks. They are 
basically by-patient numeric and text listings organized by domain and include targeted data needed for narratives. They allow for 
consolidation of information about the important domains in one place without having to look at multiple listings for the subject’s 
data. This may include patient identifier, demographic information, and other relevant information pertaining to the events of special 
interest, such as, concomitant and previous illnesses, details of timing and duration and relevant concomitant and previous 
medication. In order to get maximum benefit in writing the narratives and to quickly identify key characteristics of importance, it is 
useful to integrate numeric and text output in the PANs with graphs. The use of graphs allows for multidimensional data visualization 
and quick assessment of the relationship between data which may not be easy to pick out from text and listing format only. This 
paper will discuss the integration of numeric and text PANs with graphs using SAS. SAS code for integrating text and numeric 
output with graphs in the PANs will be illustrated. 
 
     
INTRODUCTION   
 
Establishing the safety profile of a drug is a key consideration and an important part of the drug development process. Typically, the 
primary sources of data for establishing the safety profile include safety tables, listings and graphs on AEs, laboratory data, vital 
signs, ECG, and other safety data sources. The ICH-E3 guideline calls for narratives of deaths, other serious adverse events, and 
certain other significant adverse events judged to be of special interest. To facilitate this, it is useful to provide programmed assisted 
narratives (PANs) in the form of targeted patient profiles. PANs are also useful for internal safety review, safety monitoring, and data 
integrity checks. Traditionally, PANS have been displayed in multiple tables and graphs separately and often in a disconnected 
fashion. The proposal been put forward here allows for the integration of text and graphical output allowing for multidimensional data 
visualization of a patient data and data values from multiple data domains which can aid in the assessment of patient safety. This 
paper will discuss the integration of alphanumeric PANs with graphs using SAS ODS and SAS GTL. SAS code for integrating text 
and numeric output with graphs in the PANs will be illustrated. Further, these outputs are easily customizable for specific study 
needs and hence can be used across different compounds or studies.  
  
PATIENT NARRATIVES 
 
Patient narratives are useful for conveying safety information in a concise and summary fashion. Narratives are a requirement for 
addressing safety concern of interest at the patient level. They are useful for looking at subject-specific data and for helping to 
provide some context around a specific safety issue as it pertains to individual subjects where the safety issues applies. They are 
also useful in helping to rule out and/or confirm certain safety characteristics or patterns and implications therein. Without 
programming support, preparing narratives is a labor intensive exercise (Figure 1). After statistical report (such as CSR) is 
generated, safety issues or concerns are identified which requires further investigation. Medical writers will typically add narrative 
information into the original statistical report and this gets further evaluation from Pharmacovigilance (PV). Preparing narratives 
usually require aggregation of information from different listings for each subject and summarizing in statements about the subjects 
experience regarding the event of interest. Per the ICH-E3 guideline, narratives of deaths, other SAEs, and other significant adverse 
events judged to be of special interest, such as hepatotoxicity events, renal safety events, cardiotoxicity, torsades de pointe, and so 
on should be provided. To quote ICH-E3 - 12.3.2 Narratives of Deaths, Other Serious Adverse Events, and Certain Other Significant 
Adverse Events:  
 
“There should be a brief narrative describing each death, other serious adverse event, and other significant adverse event that is 
judged to be of special interest because of clinical importance. These narratives can be placed either in the text of the report or in 
section 14.3.3, depending on their number. Events that were clearly unrelated to the test drug/investigational product may be 
omitted or described very briefly. In general, the narrative should describe the following: The nature and intensity of event; the 
clinical course leading up to event, with an indication of timing relevant to test drug/investigational product administration; relevant 
laboratory measurements; whether the drug was stopped, and when; countermeasures; post-mortem findings; investigator's opinion 
on causality and sponsor's opinion on causality, if appropriate”. 
 
  



Figure 1: A typical process for producing narratives 

 
 
Generally, the narrative will require patient identifier, demographic information, and other relevant information pertaining to the 
events of special interest, such as, concomitant and previous illnesses, details of timing and duration and relevant concomitant and 
previous medication. Before PAN was implemented, medical writing and/or PV will usually take source information and write out a 
narrative for the safety event of interest for each subject where it applies. As an example, a narrative may sound something like: 
 
 
 “A 100-year old male patient who is a current smoker with a history of hypertension, stroke, and chronic asthma was randomized to 
drug XX mg on 05-12-2012. The subject experienced an MI on Day 14 after start of taking drug XX. Concomitant medications at the 
time of the event were acenocumarol, metformin, bicalutamide, atorvastatin, and persantin. The subject was admitted to hospital on 
same day of the event. It was reported that he had first experienced worsening symptoms in late April 2012. The event was reported 
as resolved on 05-19-2012 and the subject was discharged on the same day. Laboratory results were also reported as normal at 
discharge. No change was made to the study medication dose. The investigator considered the event was unrelated to study 
medication.” 
 
As can be noted from the above example, the narrative draws from several data sources, including demographics data, drug 
dispensing data, concomitant medications, medical history data, adverse events, laboratory data, and CIOMs. A well-written 
narrative should present the information chronologically and medically make sense while at the same time identifying and relating 
the relevant information of interest in a consistent way. If the description, in terms of format, of the narrative for each subject is 
identical across all subjects, then it is possible to create programs that produce the narratives using SAS. This approach was 
discussed, for example, by Izzard and Simms (2011) and York and Martin (2009).  
 
PROGRAMMED ASSISTED NARRATIVES (PANs) 
To ease the burden of going through thousands pages of listing output, one can make use of programmed assisted narratives 
(PANs). PANs can be used to provide information that is useful in putting together narratives in the form of targeted and focused 
patient profiles. They are by-patient numeric and text listings organized by domain and include targeted data needed for narratives 
addressing a specific issue. They allow for consolidation of information about the important domains in one place without having to 
look at multiple listings for the subject’s data. Generally, the first step required for a more targeted data review is to identify the exact 
needs of the end user and ensure that all pertinent data are included. The information in PANs could include patient identifier, 
demographic information, and other relevant information pertaining to the adverse events of special interest, e.g., based on an 
SMQ, specific concomitant and previous illnesses, details of timing and duration and relevant concomitant/previous medication with 
details of dosage, and special labs if applicable.  A typical output for a PAN has a cover page as shown below (Figure 2.1). 
  



Figure 2.1 Cover Page of PAN 
 

 
 
The cover page includes a list of subjects who met the identified issues (such as AE of special interests, a set of lab criteria, etc…). 
For each subject listed, an individual narrative output file is generated using SAS program for PANas shown in Figure 2.2-2.4 below:  
 
Figure 2.2 Demographic, Study Drug Administration, Medical History and Medication History Information 
 

 
 
Figure 2.3 Adverse Events of Special Interests 
 

 
 
 
The ODS section of the program used to generate these output is included in Appendix A. The program calls a customized template 
“panrtf” which was defined using PROC TEMPLATE. The ODS section of the program are defined into two sections. The fist section 
is for the cover page shown in Figure 2.1. The second section calls a macro to automatically output one PAN file for each subject 
listed in the cover page. Within that section, the ODS options “newfile=none” and “startpage” are used to control when a page break 
should be inserted. 
  



Figure 2.4 MedicationTaken Within a Defined Time Frame 
 

 
 
 
USE OF GRAPHS TO SUPPORT NARRATIVE WRITING 

Of note from Figure 2.1-2.4, PANs in themselves are useful in the sense that only key focused information is included. Hence, the 
end user can quickly have all the information needed available readily without having to plough through pages and pages of line 
listings or non-focused output. However, it is not easy to quickly capture or identify data relationships from the PANs based on 
alphanumerical output alone. Hence writing narratives more fluidly and easily can benefit from the use of graphical output. As the 
saying goes, “a picture is worth a thousand words”. Graphical output to support alphanumeric text output has become increasingly 
important. For this purpose, it is useful to supplement PANs with graphical output which helps to quickly provide some insight into 
various relationships in the data, such as, which AEs were reported at the same time with the certain concomitant medications, or 
elevations of liver enzymes. 
 
Figure 3 – Graphical PAN 
 

 
 
 
 In Figure 3, AEs (green lines and texts), lab data (bottom part of graph), and concomitant medications (red lines and texts) can be 
presented on the same timeline taking advantage of color schemes to point out key aspects of the data characteristics and 
relationships. One can also use the graph to see the relationship between laboratory measurements of interest to see if they trend 
together; if a specific medication was given following the occurrence of specific adverse event of interest, or if a specific lab test 
value changed shortly after exposure to study drug or other drugs. Clearly a comprehensive visualization of various data from 



various domains can also help the end user to efficiently identify potential safety issues and hence from this point of view the 
graphical presentations offer an obvious advantage over the traditional alphanumeric text output as it is not easy to visualize the 
timing of events in this format.  
         
Figure 4: Integrated Alphanumeric and Graphical PAN 

 



 

 
 
 
INTEGRATION OF TEXT AND NUMERIC OUTPUT WITH GRAPHS IN PANs 
 
As noted already, for PANs with text output alone, it is not easy to readily see the relationships between the various data sources. 
On the other hand, an entirely graphical output does not render itself useful from writing a narrative without providing data. An 
optimal solution is to create PANs that combine graphical output and data. This can help writing of narratives more efficient and 
allowing the end user to quickly spot data relationships.  An example of a PAN that integrates alphanumeric and graphical output is 
provided in Figure 4.  
 
The integration of graphical output in the PANs allows the end user to quickly identify data relationships and use this along with 
specific details of data values available in the alphanumeric output when writing the narrative, all in one source.  The sample SAS 
program used to generate this kind of output is included in Appendix B. The first section of code can also be used to generate 
graphical only output as shown in Figure 3. In this section, the graphics are defined in GTL template “labplot”.  Subject information 
were output with “entry” commands. Using blockplot, the study drug periods can be identified with different color panels.  Adverse 
Events of special interests (in this case, AE related to LFT) and Medications taken within two weeks of these AEs are plotted with 
green and red colors, respectively.  The ratios of lab values relative to the upper normal limits (ULN) are presented at the bottom 
part of the graph. It makes possible to plot multiple lab parameters within a single scale by using the ratio of lab value relative to 
ULN.  However, it is not easy to identify individual lab values in it’s original unit, so as well detailed AE information.  Therefore, the 
second section of program produces two tables (lab and AE) using PROC REPORT procedures to provide detail data related to the 
graph.  Using this approach helps to provide the information needed for preparing narratives available in one source.  The same 
program can also be easily modified to plot other lab or vital sign parameters based other special safety concerns.   
 
CONCLUSION 
 
In this paper we tried to make a couple of arguments, namely, PANs can be a useful resource for writing narratives. They allow for 
consolidation of key information that is needed for narratives be available in one place. The PANs are not only useful in meeting 
operational objectives of writing narratives and aid in the creation of the submission documents, but they are also useful in 
assessing the safety profile of drugs. This way, the end user can look at information about all important data domains and 
characteristics in one place without having to look at multiple tables and graphs. Another point brought forward is that one can 
supplement PANs with graphs to help readily identify data relationship and hence also make writing of narratives more efficient. 
More importantly, we showed that one can integrate alphanumeric text with graphs in the PANs so as to maximize their value as 
resource for writing narratives and also potentially use them for submission purposes. The PANs when integrated with graphs can 
be highly customized to the needs of the users. We also illustrated how this can easily be done within the SAS environment using 
SAS ODS and SAS GTL.  In this regard, SAS posses the tools to efficiently provide customizable PAN outputs that allow the end 
user to have all the information in one place and to quickly and easily visualize and identify the key characteristics of subjects’ data 
witt respect to a specific safety concern of interest. 
 

REFERENCES 
 
Chang, S. (2005). Patient Profile, a Simple Approach. WUSS. 
 
Conover, W. (2011). Creating Hyperlinked PDF Graphical Patient Profiles with PROC REPORT. PharmaSUG2011 - Paper TU01. 
 
Light, S., Gilbert, P. and Genereux, G., (2001). A Novel Approach to Developing a Patient Profile Reporting Application 
SUGI 26, Paper P042-26. 
 
Ganesan, R., Sanjee, S.R., and Clark, J. (2010). Multidimensional Data Visualization for Generating Patient Profile 
PharmaSUG2010 - Paper AD20 
 
Huang, Y. (2007). A Low Cost Graphical Patient Profiler with SAS MetaViewApplet. 2007 PharmaSUG, Paper AD03. 
 
ICH E3 - Guideline for Industry - Structure and Content of Clinical Study Reports. 
 
Izzard, D.C. and Simms, E. M. (2004). Automation of Patient Narratives Using SAS ® and Analysis Datasets. 2004 PharmaSUG, 
Paper FC03 
 
Desgrousilliers, M. and Miralles, R. M. (2010). Creating a Patient Profile using CDISC SDTM. WUSS 2010.. 
 
Lassman, D. (2004). Automating the Production of Patient Profiles Using PROC DOCUMENT and ODS Layout 
NESUG 2004, Paper PR06. 
 
Lassman, D. (2005). Producing Patient Profiles Using PROC DOCUMENT and ODS Layout. PharmaSUG 2005, Paper CC24. 
 
Lassman, D. (2001). %ICS – A SAS® Macro to Automate Individual Case Summaries. PharmaSUG  2001, Paper CC11. 
 
Li, S., Li, D., Sproule, S., and Zhu, D. (2009). A Simple Method to Generate Patient Profiles. PharmaSUG 2009, Paper PO02. 



 
Matange, S. (2011). Tips and Tricks for Clinical Graphs using ODS Graphics. PharmaSUG 2011, Paper SAS-AD01. 
 
O’Connor, D.  and Matange, S. (2010). Create Comprehensive Patient Profiles with SAS. PharmaSug 2010, Paper CC-SAS01. 
 
Peterson, T. and Ramalingam, S. (2010). Operational Uses of Patient Profiles … Even in an eCTD and SDTM World.   
PharmaSUG 2010, Paper AD21. 
 
Silva, G. (2000). Subject Profiles: (Almost) as easy as 1, 2, 3. PharmaSUG 2000, Paper TT05. 
 
Shu, H. and Zhuge, Y. (2005). Use ODS Generating Patient Profiles. PharmaSUG 2005, Paper TT09. 
 
Tran, D.  (2008). A Novel Approach to Patient Profiling. PSI-EFSPI 2008. 
 
Typou, M. and Kim, J. (2008). A Graphical Approach to Examine the Completeness of Epidemiological Data: Patient Profile Plots 
(PPPs) and Centre Profile Plots (CPPs). PhUSE 2008. Paper TS09. 
 
York, A. and Martin, J. (2009). Using SAS to Automate the Production of Medical Writing Narratives. Phuse, 2009. 
 
Ritter, A. (2011). Creating Customized Patient Profiles using SAS ODS RTF and PROC TEMPLATE PharmaSUG, 2011, Paper 
TT03. 
 
Zheng, L. (2006). A New Approach to Display Multiple Graphs in One Page And Individual Patient Profile Graph. PharmaSUG 2006, 
Paper CC09. 
 
CONTACT INFORMATION  
Your comments and questions are valued and encouraged. Contact the author at:  
 
Faye Yeh, MS 
Takeda Global Research and Development 
One Takeda Drive 
Deerfield. Illinois 60015 
Work Phone: 224-554-2733  
E-mail: faye.yeh@takeda.com  
Web:  www.tgrd.com 

Melvin S. Munsaka, PhD  
Takeda Global Research and Development 
One Takeda Drive 
Deerfield. Illinois 60015 
Work Phone: 224-554-2846  
E-mail: Melvin.munsaka@takeda.com 
Web:  www.tgrd.com 

 
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS  
Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their 
respective companies 

http://www.tgrd.com/�
http://www.tgrd.com/�


APPENDIX A: SAS Code for PAN with Alphanumeric Output  

 
/**  Reset  **/ 
ods escapechar="~"; 
 
title; 
footnote; 
ods listing close; 
 
/**  Setup General titles and footnotes  **/  
 **  J= means justify, L for left, C for center, R for right; 
 **  ~ is the ODS escape character defined above; 
 **  {THISPAGE} and {LASTPAGE} are RTF-specific codes for producing pagination 
 **  fields in Word; 
 
title1 j=l "&COMPANY" j=r "Page ~{thispage} of ~{lastpage}"; 
title2 j=l "&protocol" ; 
title4 j=c "Information for Subjects with Markedly Abnormal LFT Values"; 
 
footnote1 j=r "~S={protectspecialchars=off pretext='\brdrt\brdrs\brdrw11 '}"; 
footnote2 j=l "&RUNINFO" ; 
 
 
/**  Cover Page  **/ 
filename _cover_ "&OUTPATH/phsugi/coverpage.rtf"; 
 
ods rtf file=_cover_ style=panrtf ; 
 
title5 j=c "Cover Page"; 
 
proc report data = AE1 nowd split='^' nocenter; 
  column SITE SUBJID TRTC1 PT_NAME AERELN2C ; 
  define SITE     /order   style(column)={just=left cellwidth=1 in} 
                           '~R"\ul Site  " '; 
  define SUBJID   /order   style(column)={just=left cellwidth=1 in} 
                           '~R"\ul Subject  " '; 
  define TRTC1    /order   style(column)={just=left cellwidth=2 in} 
                           '~R"\ul Initial Treatment  " '; 
  define PT_NAME  /display style(column)={just=left cellwidth=3 in cellspacing=0.3 in 
                                          pretext='pnhang\fi-200\li220' 
                                         } 
                           '~R"\ul MedDRA Preferred Term " '; 
  define AERELN2C /display style(column)={just=left cellwidth=1.5 in} 
                           '~R"\ul Relationship to Study drug" '; 
run; 
 
title5; 
/**  End of Cover Page  **/ 
 
*** MACRO RUNTAB(x) will  generate one output file for each subject  ***; 
%macro runtab; 
 
   proc sql; 
     reset noprint; 
     select compress(put(count(distinct &PT), 2.)) into :NTOT 
     from PANPT 
    ; 
   quit; 
 
   data BYSUB; 
     set PANPT; 
     SUBJORD = _n_; 
   run; 
 
   %do i = 1 %to &NTOT; 
 
 
    ** Output Unique Subject Identifiers into the macro variable [SUBJID}; 
     **(Used in defining patient-specific Filename below); 
     data _null_; 
       set BYSUB; 
       where SUBJORD=&i; 
       call symput('SUBJID', SUBJID); 
       call symput("SITE", SITE); 
     run; 
 
 
     filename _rtf_ "&OUTPATH/phsugi/%sysfunc(compress(&SUBJID)).rtf"; 
 
     ods rtf file=_rtf_ newfile=none nokeepn startpage=no record_separator=none 
         style=panrtf ; 
 



     title6 j=l "Subject : &SUBJID   Site: &SITE"; 
 
     /**  Subject info - 1st Part of PAN  **/ 
     proc report data = PT nowd split='^' nocenter; 
       column STUDY AGESC RACETH WEIGHT D_BMI SMOKERC; 
       define STUDY    /display style(column)={just=left cellwidth=1.2 in} 
                                '~R"\ul Study Number " '; 
       define AGESC    /display style(column)={just=left cellwidth=1.2 in} 
                                '~R"\ul Age/Gender" '; 
       define RACETH   /display style(column)={just=left cellwidth=4 in} 
                                '~R"\ul Race/Ethnicity" '; 
       define WEIGHT   /display style(column)={just=left cellwidth=1 in} format=5.1 
                                '~R"\ul Weigth(kg)" '; 
       define D_BMI    /display style(column)={just=left cellwidth=1 in} format=5.1 
                                '~R"\ul BMI(kg/m"~{super 2})'; 
       define SMOKERC  /display style(column)={just=left cellwidth=1.5 in} 
                                '~R"\ul Smoking History" '; 
       where SUBJID=&SUBJID; 
     run; 
 
     proc report data = DOSE nowd split='^' nocenter; 
       column PRDC SDT EDT DURC; 
       define PRDC   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Study Drug Regimen " '; 
       define SDT    /display style(column)={just=left cellwidth=1.5 in} 
                                '~R"\ul Start Date" '; 
       define EDT    /display style(column)={just=left cellwidth=1.5 in} 
                                '~R"\ul End Date" '; 
       define DURC   /display style(column)={just=left cellwidth=2.5 in} 
                                '~R"\ul Treatment Duration " '; 
       where SUBJID=&SUBJID; 
     run; 
 
     proc report data = PT nowd split='^' nocenter; 
       column D_LASVDT D_REASOC; 
       define D_LASVDT /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Last Visit Date " '; 
       define D_REASOC /display style(column)={just=left cellwidth=1.5 in} 
                                '~R"\ul Reason(s) for withdrawal" '; 
       where SUBJID=&SUBJID; 
     run; 
 
     proc report data = MEDHIS nowd split='^' nocenter; 
       column ALLTERM ; 
       define ALLTERM  /display style(column)={just=left cellwidth=100%} 
                                'Medical History (Stop Date) '; 
       where SUBJID=&SUBJID; 
     run; 
 
     proc report data = CONDIS nowd split='^' nocenter; 
       column ALLTERM; 
       define ALLTERM  /display style(column)={just=left cellwidth=100%} 
                                'Concurrent Conditions (Start Date) '; 
       where SUBJID=&SUBJID; 
     run; 
 
 
    /**  AE - 2nd part of PAN **/ 
    data AE; 
      set AE1; 
       where SUBJID=&SUBJID; 
      CNT = _N_; 
      MYPAGE = 1; 
      /** To avoid page overflow, only include 13 AE records in each page  **/ 
      if CNT > 13 then do; 
        CNT = 1; 
        MYPAGE + 1; 
      end; 
    run; 
 
 
     title6 j=l "Subject : &SUBJID  (Continued)"; 
     title7 j=l "Adverse Event:"; 
 
     ODS RTF startpage=now; 
     proc report data=AE  nowd split='^' nocenter; 
       column MYPAGE PT_NAME AEVT AEFREQC AEINTEC AERELN2C AEACTC AESDAY AEEDAY DURC AEOUTC AESERC ; 
       define MYPAGE   /order order = data noprint; 
       define PT_NAME  /display style(column)={just=left cellwidth=3 in cellspacing=0.3 in 
                                               pretext='pnhang\fi-200\li220' 
                                              } 
                                '~R"\ul MedDRA Preferred Term " '; 
       define AEVT     /display style(column)={just=left cellwidth=3 in cellspacing=1 



                                               pretext='pnhang\fi-200\li220' 
                                              } 
                                '~R"\ul Investigator Description" '; 
       define AEFREQC  /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Frequency " '; 
       define AEINTEC  /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Intensity" '; 
       define AERELN2C /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Relationship to Study Drug " '; 
       define AEACTC   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Action Taken" '; 
       define AESDAY   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Event Onset Date"^ (Study Day) '; 
       define AEEDAY   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Event Stop Date"^ (Study Day) '; 
       define DURC     /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Duration of Event" '; 
       define AEOUTC   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Outcome" '; 
       define AESERC   /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Serious" ^Adverse Event?'; 
 
       break after MYPAGE /page; 
     run; 
 
     /**  Concurrent Meds - 3rd Part of PAN  **/ 
     title7 j=l '~R"\ul Medications Taken Within 2 Weeks Prior to Adverse Event Onset:" '; 
 
     ODS RTF startpage=now; 
 
     proc report data=CONMED  nowd split='^' nocenter; 
       column PREFNAM OMSDAY OMEDAY CMAEC ; 
       define PREFNAM  /display style(column)={just=left cellwidth=4 in} 
                                '~R"\ul Medication Name " '; 
       define OMSDAY   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Start Date (Study Day)" '; 
       define OMEDAY   /display style(column)={just=left cellwidth=2 in} 
                                '~R"\ul Stop Date (Study Day)" '; 
       define CMAEC    /display style(column)={just=left cellwidth=1 in} 
                                '~R"\ul Taken for AE?" '; 
       where SUBJID=&SUBJID; 
     run; 
 
   ods rtf close; 
   ods listing; 
 
   %end; 
 
%mend runtab; 
 
%runtab; 
 

APPENDIX B: SAS Code for Integrating Alphanumeric and Graphical Output 

 
/**  Reset  **/ 
ods escapechar="~"; 
title; 
footnote; 
 
************************************************* 
**  Section 1: Create graphical output         ** 
************************************************* 
 
/**  Create customized GTL template “labplot”  **/  
proc template; 
  define statgraph labplot; 
    /**  Define macro variables for Demographic info  **/  
    mvar MSUBJ MSITE MAGE MSEX MRACE MHT MWT MBMI; 
    /**  Define macro variables for x and y axis range and ticks  **/ 
    nmvar XMIN XMAX YMAX YTCK; 
    begingraph; 
      entrytitle "LFT vs DOSE"; 
      layout lattice  / columns=1 rows=2 rowweights=(0.30 0.70) 
                        rowgutter=2 
      ; 
      /**  row 1 table – Subject information **/ 
      layout lattice / columns = 3 columnweights=(0.2 0.2 0.6) 
            autoalign=(topleft topright) border=true 
            opaque=true backgroundcolor=white; 
            entry halign=left textattrs=(weight=bold color=blue) "Subject: " textattrs=(weight=normal) MSUBJ ; 



              entry halign=left textattrs=(weight=bold color=blue) "Site: " textattrs=(weight=normal) MSITE; 
              entry halign=left " "; 
              entry halign=left textattrs=(weight=bold color=blue) "Age (yr)"; 
              entry halign=left textattrs=(weight=bold color=blue) "Gender"; 
              entry halign=left textattrs=(weight=bold color=blue) "Race"; 
              entry halign=left MAGE; 
              entry halign=left MSEX; 
              entry halign=left MRACE; 
              entry halign=left textattrs=(weight=bold color=blue) "Height (cm)"; 
              entry halign=left textattrs=(weight=bold color=blue) "Weight (kg)"; 
              entry halign=left textattrs=(weight=bold color=blue) "BMI (kg/m" {sup "2"} ")"; 
              entry halign=left MHT; 
              entry halign=left MWT; 
              entry halign=left MBMI; 
 
            endlayout; 
 
      /** row 2: graph  **/ 
        layout overlay /xaxisopts=(linearopts=(tickvaluelist=(-28 1 14 28 42 56 70 84 98 112 126) 
                                   viewmin=XMIN viewmax=XMAX) 
                                   offsetmin=0 offsetmax=0 
                                   ) 
                        yaxisopts=(linearopts=(tickvaluelist=(0 2 3 5 10) 
                                   viewmin=0 viewmax=YMAX ) 
                                   offsetmin=0 offsetmax=0.1 
                                   ); 
         blockplot x=D_SDAYSP block=TRTC / valuevalign=top 
         datatransparency=.3 valuefitpolicy=shrink name="TRTNAME"; 
          /**  AE terms  **/ 
          vectorplot x=D_EDAYSP y = YAE xorigin= D_SDAYSP yorigin = YAE 
             / arrowheads=false lineattrs=(color=green); 
 
          scatterplot x=XTM y = YAE / markercharacter=TERM markercharacterattrs=(color=green SIZE=6); 
          scatterplot x=D_SDAYSP y = YAE / markerattrs=(color=green size=6px symbol=trianglefilled); 
          scatterplot x=D_EDAYSP2 y = YAE / markerattrs=(color=green size=6px symbol=trianglefilled); 
 
          /**  OM terms  **/ 
          vectorplot x=D_EDAYSP y = YOM xorigin= D_SDAYSP yorigin = YOM 
            / arrowheads=false lineattrs=(color=red); 
 
          scatterplot x=XTM y = YOM / markercharacter=TERM markercharacterattrs=(color=red SIZE=4); 
          scatterplot x=D_SDAYSP2 y = YOM / markerattrs=(color=red size=6px symbol=trianglefilled); 
          scatterplot x=D_EDAYSP2 y = YOM / markerattrs=(color=red size=6px symbol=trianglefilled); 
 
          /**  lab  **/ 
          seriesplot x=D_SDAYSP y=VALUE /primary=true display=all markerattrs=(size=4px) 
                                         group=LTSTABR name="TEST"; 
                                         referenceline y=2 / lineattrs=(pattern=dot thickness=2); 
          discretelegend "TRTNAME" / location=inside autoalign=(topleft) 
                                     opaque=true across=2 pad=(right=5); 
          discretelegend "TEST"; 
        endlayout; 
      endlayout; 
    endgraph; 
  end; 
run; 
 
 
/**  Output graphical portion PAN using SGRENDER  **/ 
ods listing close; 
 
filename _rtf_ "&OUTPATH/phsugi/&PROG..rtf"; 
 
options orientation = portrait; 
 
ods rtf file=_rtf_  style=pan_grf startpage=never; 
 
ods layout start columns=1 rows=2; 
 
ods region width=6.5 in ; 
ods graphics on / width=6.5 in ; 
proc sgrender data=FINAL template=labplot; 
run; 
 
ods graphics off; 
 
/**  Section 2: Generate alphanumerical output  **/ 
ods rtf style=intext startpage=never; 
ods region width=8.5 in ; 
 
/**  Lab table  **/ 
proc report data = TBLRPT nowd nocenter split='^' 
  style(report)={font_size=8 pt leftmargin=1 in } 



; 
  column D_SDAYSP DAY LAB104 LAB105 LAB106 LAB107 LAB108; 
  define D_SDAYSP/ order noprint; 
  define DAY     / style(column)={just=left cellwidth=1.2 in} 
                   "Sample Date^(Study Day)"; 
  define LAB104  / style(column)={just=left cellwidth=1.2 in} 
                   "Total^Bilirubin^(3-21 umol/L)"; 
  define LAB105  / style(column)={just=left cellwidth=1 in} 
                   "Alkaline^Phosphatase^(31-106 U/L)"; 
  define LAB106  / style(column)={just=left cellwidth=1 in} 
                   "AST^(SGOT)^(9-34 U/L)"; 
  define LAB107  / style(column)={just=left cellwidth=1 in} 
                   "ALT^(SGPT)^(9-34 U/L)"; 
  define LAB108  / style(column)={just=left cellwidth=1 in} 
                   "GGT^ (4-49 U/L)"; 
run; 
 
/**  AE table  **/ 
proc report data = AETBL nowd nocenter split='^' 
  style(report)={font_size=8 pt leftmargin=1 in } 
; 
 
  column AEORD TERM AESDAY AEEDAY AEINTEC AERELN2C SAE AEACTC; 
  define AEORD   / order noprint; 
  define TERM    / style(column)={just=left cellwidth=2 in} 
                   "AE Preferred Term"; 
  define AESDAY  / style(column)={just=left cellwidth=1 in} 
                   "AE onset Date^(Study Day)"; 
  define AEEDAY  / style(column)={just=left cellwidth=1 in} 
                   "AE End Date^(Study Day)"; 
  define AEINTEC / style(column)={just=left cellwidth=0.5 in} 
                   "Intensity"; 
  define AERELN2C/ style(column)={just=left cellwidth=1 in} 
                   "Relationship^to^Study Drug"; 
  define SAE     / style(column)={just=left cellwidth=1 in} 
                   "Serious AE^/Death Date"; 
  define AEACTC  / style(column)={just=left cellwidth=0.8 in} 
                   "Action Taken"; 
run; 
 
ods layout end; 
 
ods rtf close; 
ods listing; 
 
 

 


	Establishing the safety profile of a drug is a key consideration and an important part of the drug development process. Typically, the primary sources of data for establishing the safety profile include safety tables, listings and graphs on AEs, labor...

