
ABSTRACT 
Do you know how to create titles for multiple clinical reports directly from data using SAS PROC FORMAT within a 
Do loop? When we work with clinical efficacy data, we often have the total score and sub item score. Sometimes, the 
efficacy data contains up to 50+ sub items. Each item and total score will go through the same analysis and get 
outputs from the statistical models. Then, tables are put into one file with one item per page. Traditionally, we create 
macro using hard coding via macro parameter for the title of the parameter description in each table. However, there 
is a more efficient way to do this. We can use proc format from the data to create the title of the parameter description 
for each item table automatically within a Do loop. Using this method, we can avoid hard coding or possible typos in 
the titles. This paper introduces how to utilize this tip and trick. 

 
INTRODUCTION 
Why do we need to avoid hard coding to create titles for multiple reports?  

When we create only one report or graph, it is easy for us to just type the title before PROC REPORT or PROC 
GPLOTS statement. But in the NDA submission, we often are asked to create multiple reports repeatedly for different 
parameters that go through the same statistical analysis. In those situations, we need to create a title for every 
parameter in each table or graph. We can accomplish this by typing the title for each table, but this takes a lot of time 
to complete. Additionally, the potential for typos is higher because there are many titles to type.  

 

 

PARAMETER_CD PARAMETER_Details STAT TRT1 TRT2 VAR_ORD

ITEM1 AAAAAA n 102 102 1 

ITEM1 AAAAAA lsm 0.23 0.40 2 

ITEM1 AAAAAA ci (0.15, 0.31) (0.32,0.48) 4 

ITEM1 AAAAAA se 0.05 0.05 5 

ITEM2 BBBBBB n 102 102 1 

ITEM2 BBBBBB lsm 0.23 0.40 2 

ITEM2 BBBBBB ci (0.15, 0.31) (0.32, 0.48) 4 

ITEM2 BBBBBB se 0.05 0.05 5 

…………… ……………. ……… …… …… … 

ITEM9 FFFFFFF n xxx xxx 1 

ITEM9 FFFFFFF lsm x.xx x.xx 2 

ITEM9 FFFFFFF ci (x.xx-x.xx) (x.xx-x.xx) 4 

ITEM9 FFFFFFF se xx.x xx.x 5 

Table1.  Example data FINAL 

An efficient method to create titles for multiple clinical reports using proc format within a Do loop, continued 

 

THE TYPING METHOD FOR THE REPORT TITLES BY USING MACRO PARAMTERS 
 
 
***********************************************; 
** macro parameters:                         **; 
** datain - Input data                       **; 
** title  - The title for the parameter      **; 
** item   - Item code                        **; 
***********************************************; 
 
%macro report(datain=, title=, item=); 
data report;  
set &datain;     
  where PARAMETER_CD =”&item”; 
run; 
 
title1 “Summary of Mean Change from Baseline in Treatment Period”; 
title2 “Efficacy parameter: &title”; 
 
PROC REPORT DATA=report HEADLINE HEADSKIP SPLIT = '#' ; 
  COLUMNS ( '--'  var_ord  ("Actual treatment Period" "--" TRT1 TRT2)  ); 
  …………………………………; 
Run; 
%mend report; 
 
%report(datain=final, title=AAAAAA, item=ITEM1); 
%report(datain=final, title=BBBBBB, item=ITEM2); 
…………………………………………………………; 
…………………………………………………………; 
%report(datain=final, title=FFFFFFF, item=ITEM9; 
 
 

AN EFFICIENT METHOD TO CREATE TITLES FOR MULTIPLE REPORTS OR GRAPHS 
When we work on multiple parameters with similar outputs, each parameter will be put in a table, and the parameter 
description should be created for each table at the title. This method of creating title for each report is more efficient, 
and also eliminates the potential for typos. This method includes four steps: 

 
STEP 1: CREATE DESCRIPTION OF PARAMETER USING PROC FREQ FROM DATA 
Most clinical trial data contains variable parameter code and details. Each parameter code has a descriptive variable 
to explain the content of parameter (e.g. efficacy item code with item content, or lab test code with lab test name, 
etc.)  
 

We use table 1 data as our example. There are variables PARAMETER_CD and PARAMETER_Details in the data. 
We can use PROC FREQ to create a dataset called PARA_DATA, and then create a numeric order variable for each 
parameter: 

 

Proc freq data=final noprint; 
  Tables PARAMETER_CD*PARAMETER_Details /out=para_data(keep= PARAMETER_CD   
PARAMETER_Details); 
 
Proc sort data=para_data;  
  by PARAMETER_CD PARAMETER_Details; 
run;  

An efficient method to create titles for multiple clinical reports using proc format within a Do loop, continued 

 

 
 
********************************************************; 
*** Create a numeric order variable for each parameter**; 
********************************************************; 

 

Data para_data;  
set para_data; 
  PARAMETER_order=_n_; 
Run; 

 

 
PARAMETER_CD PARAMETER_Details PARAMETER_order
ITEM1 AAAAAA 1 
ITEM2 BBBBBB 2 
ITEM3 CCCCCC 3 
ITEM4 DDDDDD 4 
ITEM5 EEEEEE 5 
ITEM6 FFFFFFF 6 
ITEM7 GGGGG 7 
ITEM8 HHHHHH 8 
ITEM9 FFFFFFF 9 

Table 2.  Data  PARA_DATA 

 
 

SETP 2: THE FORMAT DATA SHOULD CONTAIN VARIABLE START, LABEL, AND FMTNAME 
The data that is put into proc format should contain variable START, LABEL, and FMTNAME. The codes below 
create the FORMAT data from DATA  PARA_DATA. 

 

data itemlnm;  
set para_data; 
  length label $50.; 
  start=PARAMETER_order; 
  label=PARAMETER_Details; 
  Fmtname='itemn'; 
  keep start label fmtname; 
run; 
 

 

STEP 3: USE THE OPTION CNTLIN IN PROC FORMAT TO CREATE SAS FORMAT  
 
proc format cntlin=itemlnm  library=work; 
run;  

 

STEP4: USING A DO LOOP TO CRATE MULTIPLE REPORTS  
a. Merge data PARA_DATA with data FINAL by PARAMETER_CD and PARAMETER_Details 

 

data final;  
merge final para_data;  
  by PARAMETER_CD PARAMETER_Details; 
run; 

An efficient method to create titles for multiple clinical reports using proc format within a Do loop, continued 

 

b. Use the macro for the Do loop to create reports: 
 

%macro rpt; 
 
***************************************************************; 
****Create a macro variable for the maximum of item number*****; 
***************************************************************; 

 
data _null_;  
set para_data; 
  call symput('parameterNmax', trim(left(put(_n_, 6.0)))); 
  run; 
%put &parameterNmax; 

   
***************************************************************;   
***** a Do loop for the report where item from 1 to maximum ***;  
***** number of items                                       ***; 
***************************************************************; 
 
%do item_order=1 %to &parameterNmax; 
Data report&item_order;  
set final; 
  Where PARAMETER_order=&item_order; 
Run; 

 
***************************************************************;  
****Use SAS format, %SYSFUNC, and PUTN within a Do loop to  ***; 
****create a macro variable for each parameter description  ***;  
****at title for the reports                                ***; 
****The format ‘itemn’ is created from STEP 2 and 3         ***;  
***************************************************************; 

 
%let itemname=%sysfunc(putn(&item_order, itemn.)); 
%put &itemname; 
 
TITLE1 "Summary of Mean Change from Baseline in Treatment Period"; 
TITLE2 "Patient Population"; 
 
PROC REPORT DATA=report&item_order HEADLINE HEADSKIP SPLIT = '#' ; 
  COLUMNS (var_ord ("Actual Treatment Period" "--" TRT1 TRT2)); 
  DEFINE var_ord/ DISPLAY WIDTH=25 ' ' left format=varord. flow; 
  DEFINE trt1 / DISPLAY WIDTH=15  "Treatment1" left; 
  DEFINE trt2 / DISPLAY WIDTH=15  "Treatment2" left; 
 
compute before _page_; 
    line @2 130*'-'; 
    line @2 "Efficacy Parameter: &itemname"; 
endcomp; 
 
 
Run;   
%end; 

 
%mend rpt; 

 
*****************************; 
*** Invoke macro RPT      ***; 
*****************************; 
 
%rpt; 

Summary of Mean Change from Baseline in Treatment Period
Patient Population 

       
                                                                                                                             
 ------------------------------------------------------------------------------------------------------------------------------------------------------------------- 
Efficacy Parameter: AAAAA 
 
                                                                                                        Actual Treatment Period                     
                                                                                             ------------------------------------------------------------------ 
                                                                                                                           
                                                                                             Treatment1     Treatment2        
  ----------------------------------------------------------------------------------------------------------------------------------------------------------------- 
                                                                                                                                     
    Sample Size                         102                                      102          
                            Least Squares Means                    0.23                                    0.40          
                              90% Confidence Interval      (0.15, 0.31)                 (0.32, 0.48)    

                 Standard Error                        0.05                           0.05          

Output1. Report for First Item from the Macro above 

 

CONCLUSION  
Compared with the traditional method of typing the title in each output, we can create titles more efficiently and 
accurately for reports using the method introduced in this paper. The main benefits of using this method are as 
follows:  

i. Saves time: The Do loop creates the report with the title automatically. 
 

ii. Avoids typos and ensures accuracy: The title or descriptive label for the Item created from data directly. 
 

iii. Versatile and useful in many situations: Since the title is created in a macro variable, it is easy to use this 
macro variable everywhere in tables, listings or graphs. 

