
ABSTRACT 
In order to expedite the review process, we often get request from the client or business user to combine all tables 
and listings into one rtf document.  Unfortunately, SAS does not provide a function to combine multiple rtf documents. 
Also, before finalization of the study draft listings and tables are generated more frequently for reviews. Unknowingly, 
using this draft listings and tables produced in SAS® for business needs increases the risk associated with misusing 
or misinterpreting the results. A viable solution to reduce that risk is to use a visible watermark embedded in the SAS 
outputs such as a pale image or text displayed and/or printed behind the text. This paper will introduce a method for 
combining the multiple rtf documents into one rtf document and adding watermark to the final RTF document in SAS 
using word basic commands via DDE (Dynamic Data Exchange). If needed, these functions can be use 
independently. 

INTRODUCTION 
In Pharmaceuticals/CRO industries, to expedite the review process business user frequently request to combine all 
tables and listings into one rtf document. Since, rtf document can be use as a word document it is easier for them to 
review and provide comments using the track change option. Having all the comments back in one document will 
make programmer job easier and save time. This paper will introduce a method to combine multiple rtf documents in 
SAS using word basic commands via DDE. 

During the life cycle of the study customers regularly use the reports produced in SAS for business needs; if they are 
in draft, this can run the risk of misusing or misinterpreting the results. To prevent this, visible watermark such as 
word ‘Draft’ can be place into the report which will remind the readers about the report status. The visible watermark 
is a pale image or text displayed and/or printed “behind” text in the document and a valuable unobtrusive way to 
display important information that cannot be missed by the reader such as the fact that the document is a draft 
version or confidential. In SAS 9.1 or above, using ODS with a template predefined by Proc Template, an image can 
be placed behind a table or listing. This paper introduces an alternative and easier way to place a watermark in a 
report using SAS. The beauty of this solution is the flexibility to control the size, position and appearance of the 
watermark.  

This paper introduces a macro which will combine all the rtf documents from a given location and apply watermark on 
the final rtf document. The entire process involves SAS, Word, and VBA macro which is automated by a SAS macro 
called %combine_rtf_watermark. 

TECHNIQUE & MECHANISM 
The program is divided into two part first part will describe the process of combining multiple rtf documents and 
second part will describe the process of creating a watermark. 

The general process of combining multiple rtf documents are as follows: 

1. Get the list of all rtf documents from a given location using pipe command and further create macro 
variables for each file name. 

2. Open the first file insert a section break at the end of document. 
3. Insert the second file. 
4. Repeat step 2 and 3 until all documents are combined using do loop. 
5. Save final document as Master.rtf and close. 

The general process of watermarking RTF output in SAS is as follows (note that the first item is a one-time setup, and 
the rest can be managed via SAS code): 

1. Record VBA Macro in Word to create the watermark on plain document. 
2. Open the Master.rtf in Word and execute the VBA Macro using word basic command and save. 

To automate these steps, the DDE solution is leveraged to build the communication bridge between SAS and Word 
2003.  WordBasic commands (Microsoft Corporation, 1999) can be sent from SAS via DDE to enable SAS to take 
control of Word 2003. 

GETTING LIST OF RTF FILES AND CREATING MACRO VARIABLE FOR EACH FILE: 
To get the list of all rtf documents from a given location pipe command is use as follows: 

filename dirlist pipe "dir  ""&path"" /b "; 

data dirlist1 ; 
 infile dirlist length=reclen ; 
 input filename $varying1024. reclen ; 
 if scan(filename,2,'.') in ('doc' 'rtf'); 
 path="&path"; 
 filename=strip(path)||'\'||strip(filename); 
run; 

The above data step will get the lists of all the .rtf files to be combined from the given location and a variable filename 
combining the location and filename is created. 

Following code will count the total number of documents and also assign a macro variable for each document from 
the variable filename.  

proc sql noprint; 
select count(*) into:n from dirlist1; 

 select filename into:file1-:file%cmpres(&n)  
  from dirlist1; 
quit; 

The above proc step will assign a macro variable for each file and also count the total number of files in a given 
location. 

COMBINING ALL RTF DOCUMENTS USING WORD BASIC COMMAND: 
Prior to executing these statements, there are two system options needed NOXWAIT and NOXSYNC. The NOXWAIT 
option specifies that the DOS command prompt window disappears without your having to type EXIT when the 
process is finished, while the NOXSYNC specifies that the process should execute asynchronously. That is, control is 
returned immediately to the SAS System and the command continues executing without interfering with your SAS 
session. 

In order for a client/server communication link to be established, both SAS and Word must be running. Therefore, for 
the first iteration, it is necessary to programmatically launch Word from a SAS session. There are several techniques 
available to launch Word from SAS. The simplest one is the following statement: 

 %let rc=%sysfunc(system(start winword)); 

The above command is dependent on the finishing of previous command. So, SLEEP function can be used frequently 
in the SAS command/datastep that is dependent on previous job finishing. These will avoid the error occur due to 
delay in the execution of previous SAS command/datastep. 

The syntax for the SLEEP function is given below: 

 data _null_; 
  x=sleep(5); 
 run; 

Above step will pause SAS session for five seconds. 

To communicate with Word from SAS, the following FILENAME statement is used to establish the linkage between 
SAS and Word via the DDE triplet: 

 filename word DDE 'Winword|System'; 

The following data step will combine all the rtf documents into one file. This data step will initially open the first file 
and add a section break at the end of document. Once the section break is added the next file will be inserted using 
word basic command. This process will be continuing unless all files are combined. Once combined, the macro will 
save the final rtf file as Master.rtf and close the winword session. 

data _null_; 
 file word; 
 put '[FileOpen .Name = "' "&file1" '"]'; 
 %do i=2 %to &n; 
  put '[EndOfDocument]'; 
  put '[InsertBreak .Type = 2]'; 
  put '[InsertFile .Name ="' "&&file&i." '" ]'; 
 %end; 
 put '[FileSaveAs .Name ="&path\master.rtf"]'; 
 put '[FileSave]'; 
 put '[FileCloseAll]'; 
 put '[FileExit]'; 

run; 

In the above macro Fileopen command will open the first file and then cursor will move to the end of file by 
EndofDocument command. Further, InsertBreak and Insertfile command will insert a section break and 
insert the file. It is necessary to programmatically save and close the document with the FileSave and FileClose 
command. Further, FileExit command will exit from the Microsoft Word. 

RECORD VBA MACRO IN WORD TO ADD WATERMARK: 
For many, perhaps most SAS developers the easiest and fastest way to write VBA code is to first go into a Microsoft 
application (in this paper Microsoft Word is used as the application), turn on the macro recorder, perform the steps 
and functions desired, and then terminate the macro recorder. The end result is a stored VBA program (macro) that 
can be executed at a later date. This makes it possible for a SAS developer to automate tasks in the Microsoft 
application, and therefore vastly improve the functionality of an integrated system that takes advantage of the relative 
strengths of the SAS System and the Microsoft application. 

Steps to record VBA Macro are as follows: 

1. Open Word. 
2. Select Tools, Macro, Record New Macro from the pull down menus. 
3. Enter the name of macro as ‘Watermark’. 
4. Select Ok. 
5. Select Format from the standard toolbar. 
6. On the format menu, point to background, and then click Watermark. 
7. In the Printed Watermark dialog box, click the radio button next to Text watermark. User can also select the 

picture watermark. 
8. In the Text field you can either select from the list in the drop-down menu or you can type directly in the Text 

field. 
9. Select the font, size and color. 
10. Select Ok. 
11. Select the stop recording button. 
12. Close the Word. 

OPEN THE TARGET FILE IN WORD AND EXECUTE THE VBA MACRO: 
The same approach given above is used to start a new Word session: 

 %let rc=%sysfunc(system(start winword)); 

To communicate with Word from SAS, the following FILENAME statement is used to establish the linkage between 
SAS and Word via the DDE triplet: 

 filename word DDE 'Winword|System'; 

The next step is to open the target file “&in” by sending the FileOpen WordBasic command to Word with a 
data _null_ step. Once the file is open in Word, VBA macro ‘Watermark’ is executed as shown in the following 
steps: 

 data _null_; 
  file word; 
   put '[FileOpen.Name = "' "&in" '"]'; 
   put '[Watermark()]'; 
   put ‘[FileSave]'; 
   put '[FileClose]'; 
   put '[FileExit]'; 
 run; 

“&in” is the macro variable containing the name and location of target file. 
It is necessary to programmatically save and close the document with the FileSave and FileClose command. Further, 
FileExit command will exit from the Microsoft Word. 

%Combine_RTF_Watermark 
To facilitate and automate the above discussed steps from combining the RTF files and applying the watermark,  
a SAS macro called %combine_rtf_watermark was developed for SAS v9.1.3 or above (see Appendix for 
details). The user can easily extend the macro to fit other SAS versions. 

There are only two keyword parameters: 

In: Define the path and the name of the input Word file, e.g., C:\Demo\test.rtf. 

Path: Production Location, for e.g., C:\Demo. 

Below is the simple macro call to % Combine_RTF_Watermark. 

 % combine_rtf_watermark (in= C:\Demo\test.rtf, path=C:\Demo); 

Below is the example showing watermark on the master.rtf. 

 

Figure 1. Master.rtf with Watermark. 

CONCLUSION 
Above solution demonstrates an innovative approach of watermarking and combining multiple RTF outputs without 
any formatting changes in the document. This solution requires minimal amount of coding and can be further modified 
to take advantage of other Word functions to enhance the RTF output. 

APPENDIX 
%macro combine_rtf_watermark (in=,path=); 
 

options noxwait noxsync; 
 
%let in=&path\master.rtf; 
 
filename dirlist pipe "dir  ""&path"" /b "; 
 
data dirlist1 ; 
 infile dirlist length=reclen ; 
 input filename $varying1024. reclen ; 

if scan(filename,2,'.') in ('doc' 'rtf'); 
 path="&path"; 
 filename=strip(path)||'\'||strip(filename); 
run; 
 
proc sql noprint; 

select count(*) into:n from dirlist1; 
 select filename into:file1-:file%cmpres(&n)  
  from dirlist1; 
quit; 

 
%let rc=%sysfunc(system(start winword)); 

 
 data _null_; 
  x=sleep(5); 
 run; 
 
 filename word DDE 'Winword|System'; 

 
data _null_; 
 file word; 
 put '[FileOpen .Name = "' "&file1" '"]'; 
 %do i=2 %to &n; 
  put '[EndOfDocument]'; 
  put '[InsertBreak .Type = 2]'; 
  put '[InsertFile .Name ="' "&&file&i." '" ]'; 
 %end; 
 put '[FileSaveAs .Name ="&path\master.rtf"]'; 
 put '[FileSave]'; 
 put '[FileCloseAll]'; 
 put '[FileExit]'; 
run; 

 
data _null_; 

  x=sleep(5); 
 run; 
 

%let rc=%sysfunc(system(start winword)); 
 
 filename word_w DDE 'Winword|System'; 
 

 data _null_; 
  file word_w; 
   put '[FileOpen.Name = "' "&in" '"]'; 
   put '[Watermark()]'; 
   put ‘[FileSave]'; 
   put '[FileClose]'; 
   put '[FileExit]'; 
 run; 
 

%mend combine_rtf_watermark; 

